
Handle terminal events
    

Custom components can have local state. To make them becomes interactive, they need to react to terminal events. There are two ways to achieve that:

  
    
  
  Init configuration


In component's init callback, we can return a configuration with events_subscription: true to subscribe events for the component.
defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      %Orange.Terminal.KeyEvent{code: {:char, char}} ->
        IO.puts("Key pressed: #{char}")
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 20, height: 20] do
      "Hello"
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start(Example)
Your component can handle events with the handle_event/3 callback and returns the updated state.

  
    
  
  Orange.subscribe/1


If you want to manually subscribe, use Orange.subscribe/1.
defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def after_mount(_state, _attrs, _update) do
    Orange.subscribe(:root)
  end

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 20, height: 20] do
      "Hello"
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start({Example, id: :root})

  
    
  
  Remove events subscription


To unsubcribe for events, use Orange.unsubcribe/1.

  
    
  
  Component focus mode


There are certain cases that you only want one component to receive events and prevent other components. Look at the above example, an input component should receive all the events and prevent users from quitting the application when hitting "q". To do that, we can set a component as focused with Orange.focus/1. Only the focused component can receive terminal events. To remove the focus status, use Orange.unfocus/1.



  

    
Orange 
    



      
Terminal UI framework for Elixir.
A framework empowering developers to build complex terminal UI applications. Orange provides a declarative way to build UI components and manage terminal events. Orange offers stateful components which can encapsulate local state and logics, allows you to split your application into discrete sub-components. For example, here is a panel displays a random numbers every 1 second:
defmodule Panel do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: 0, events_subscription: true}

  @impl true
  def after_mount(_state, _attrs, update) do
    spawn_link(fn -> random_number(update) end)
  end

  defp random_number(update) do
    number = :rand.uniform(100)
    update.(number)
    Process.sleep(1000)
    random_number(update)
  end

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  @impl true
  def render(state, _attrs, _update) do
    rect style: [width: 25, height: 10, border: true] do
      span do
        "Random number: #{state}"
      end
    end
  end
end

# Start the application. To quit, press 'q'.
Orange.start(Panel)

      


      
        Summary


  
    Functions
  


    
      
        focus(component_id)

      


        Focus a component by component_id.



    


    
      
        start(element)

      


        Start the runtime and render the UI root.



    


    
      
        stop()

      


        Stop the runtime



    


    
      
        subscribe(component_id)

      


        Subscribe to the event manager.



    


    
      
        unfocus(component_id)

      


        Unfocus a component by component_id.



    


    
      
        unsubscribe(component_id)

      


        Unsubscribe to the event manager.



    





      


      
        Functions

        


  
    
      
      Link to this function
    
    focus(component_id)



  


  

Focus a component by component_id.
There can be only one focused component at a time. If a component is focused, it will receive all the events and prevent other components from receiving them. This is useful when you want a component to intercept all terminal events, for example, an input.

  



  
    
      
      Link to this function
    
    start(element)



  


  

Start the runtime and render the UI root.
The root component must be a custom component.

  



  
    
      
      Link to this function
    
    stop()



  


  

Stop the runtime

  



  
    
      
      Link to this function
    
    subscribe(component_id)



  


  

Subscribe to the event manager.
Subscribed components will receive terminal events. To unsubscribe, use unsubscribe/1.

  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def after_mount(_state, _attrs, _update) do
    Orange.subscribe(:root)
  end

  @impl true
  def render(_state, _attrs, _update) do
    span do
      "Hello"
    end
  end
end

Orange.start({Example, id: :root})

  



  
    
      
      Link to this function
    
    unfocus(component_id)



  


  

Unfocus a component by component_id.

  



  
    
      
      Link to this function
    
    unsubscribe(component_id)



  


  

Unsubscribe to the event manager.

  


        

      



  

    
Orange.Macro 
    



      
Macros for creating Orange primitive components
Currently, Orange supports three primitive components:
	rect for creating a rectangle
	line for creating a line
	span for creating a span


  
    
  
  Examples


Macro provides an ergonomic way to create Orange components. For example, the following code:
rect style: [width: 12, border: true], direction: :row do
  span style: [color: :red] do
    "Hello"
  end

  span do
    "World"
  end
end
will produce the following struct:
%Orange.Rect{
  children: [
    %Orange.Line{
      children: [
        %Orange.Span{children: ["Hello"], attributes: [style: [color: :red]]}
      ],
      attributes: []
    },
    %Orange.Line{
      children: [%Orange.Span{children: ["World"], attributes: []}],
      attributes: []
    }
  ],
  attributes: [style: [width: 12, border: true], direction: :row]
}

  
    
  
  Components children


Orange components can have children, provided by the do block. The syntax took inspiration from HTML:
rect do
  span do
    "Hello"
  end

  span do
    "World"
  end
end
The children can also be a list of components. This is useful when you want to render a collection of components. For example:
rect do
  [
    span do
      "Hello"
    end,

    span do
      "World"
    end
  ]
end
Normal Elixir expression works inside the children block. For example, here's how to conditionally render components:
rect do
  if :rand.uniform() > 0.5 do
    "Hello"
  else
    "World"
  end
end

  
    
  
  Styling



  
    
  
  Sizing


	:width - the width of the component

	:height - the height of the component


The values for width and height can be:
	An integer - the number of cells in the terminal. For example: style: [width: 10]

	A percentage string - the size is equal to the given percentage of the parent size. For example: style: [width: "50%"]

	An calc expression - the size is calculated based on the given expression. For example: style: [width: "calc(100% - 10)"]

	A fraction string - the size is calculated based on the given fraction of the parent size. All sibling components MUST also have a fraction size. The size will be calculated based on the component fraction divided by the sum of all sibling fractions. For example:
rect style: [width: 12] do
  rect style: [width: "1fr"] do
  end

  rect style: [width: "2fr"] do
  end
end
In this example, the first rect will have a width of 1/3 of the parent rect, and the second rect will have a width of 2/3 of the parent rect.



  
    
  
  Padding


Padding for the render box's inner content. The values for padding can be:
	One integer - padding for all sides. For example: style: [padding: 1] means [padding_top: 1, padding_bottom: 1, padding_left: 1, padding_right: 1]

	Two integers tuple - padding vertical and padding horizontal. For example: style: [padding: {1, 2}] means [padding_top: 1, padding_bottom: 1, padding_left: 2, padding_right: 2]

	Four integers tuple - padding top, right, bottom, left respectively. For example: style: [padding: {1, 2, 3, 4}] means [padding_top: 1, padding_bottom: 3, padding_left: 4, padding_right: 2]



  
    
  
  Border


	:border - whether to render a border around the rect. Defaults to false

	:border_top - whether to render a top border. Defaults to true if border is true

	:border_bottom - whether to render a bottom border. Defaults to true if border is true

	:border_left - whether to render a left border. Defaults to true if border is true

	:border_right - whether to render a right border. Defaults to true if border is true

	:border_color - the color of the border. See Color section for supported colors



  
    
  
  Color


	:color - the color of the component text. The color value can be inherited from the parent component. If the color value is not specified, the component will inherit the color from the parent component

	:background_color - the color of the component background


The values for :color and :background_color is a single atom representing the color. Supported colors are:
	:white
	:black
	:grey
	:dark_grey
	:red
	:dark_red
	:green
	:dark_green
	:yellow
	:dark_yellow
	:blue
	:dark_blue
	:magenta
	:dark_magenta
	:cyan
	:dark_cyan


  
    
  
  Position


Rect elements support positioning. Supported values are:
	{:fixed, top, right, bottom, left} - the rect will be fixed at the given position. The values are offset to the respective edge of the screen. For example:
rect position: {:fixed, 1, 2, 1, 2} do
  "Fixed position"
end
will render:
[image: rendered result]



      


      
        Summary


  
    Functions
  


    
      
        line(attrs \\ [], do_block)

      


        Generates a Orange.Line struct



    


    
      
        rect(attrs \\ [], do_block)

      


        Generates a Orange.Rect struct



    


    
      
        span(attrs \\ [], do_block)

      


        Generates a Orange.Span struct



    





      


      
        Functions

        


    

  
    
      
      Link to this macro
    
    line(attrs \\ [], do_block)



      (macro)

  


  

Generates a Orange.Line struct

  
    
  
  Options


	:style - style attributes for the line. Supported keys are:
	:width - see Sizing section

	:height - see Sizing section

	:padding - see Padding section

	:color - see Color section

	:background_color - see Color section





  
    
  
  Children validation


Line only accepts Orange.Span as children. To provide better ergonomics, the macro will automatically wrap string to Orange.Span. For example, the following code:
line do
  "Hello"
end
  will produce the result:
%Orange.Line{
  children: [%Orange.Span{children: ["Hello"], attributes: []}],
  attributes: []
}

  
    
  
  Examples


iex> import Orange.Macro
iex> line style: [width: 5, color: :red] do
...>   span do
...>     "foo"
...>   end
...> end
%Orange.Line{
  children: [%Orange.Span{children: ["foo"], attributes: []}],
  attributes: [style: [width: 5, color: :red]]
}

  



    

  
    
      
      Link to this macro
    
    rect(attrs \\ [], do_block)



      (macro)

  


  

Generates a Orange.Rect struct

  
    
  
  Options


	:direction - the direction to layout children :row or :column (default)

	:style - style attributes for the rect. Supported keys are:
	:width - see Sizing section

	:height - see Sizing section

	:border - see Border section

	:padding - see Padding section

	:color - see Color section

	:background_color - see Color section



	:title - the title of the rect. If specified, it implies border is true

	:scroll_x - the horizontal scroll offset

	:scroll_y - the vertical scroll offset

	:position - the position of the rect. See Position section



  
    
  
  Children validation


Rect only accepts Orange.Rect and Orange.Line as children. To provide better ergonomics, the macro will automatically wrap Orange.Span and string to Orange.Line. For example, the following code:
rect do
  "Hello"
end

rect do
  span do
    "Hello"
  end
end
  will both produce the result:
%Orange.Rect{
  children: [
    %Orange.Line{
      children: [%Orange.Span{children: ["Hello"], attributes: []}],
      attributes: []
    }
  ],
  attributes: []
}

  
    
  
  Examples


iex> import Orange.Macro
iex> rect style: [width: 5, border: true], direction: :row do
...>   "foo"
...>
...>   span do
...>     "bar"
...>   end
...> end
%Orange.Rect{
  children: [
    %Orange.Line{
      children: [%Orange.Span{children: ["foo"], attributes: []}],
      attributes: []
    },
    %Orange.Line{
      children: [%Orange.Span{children: ["bar"], attributes: []}],
      attributes: []
    }
  ],
  attributes: [style: [width: 5, border: true], direction: :row]
}

  



    

  
    
      
      Link to this macro
    
    span(attrs \\ [], do_block)



      (macro)

  


  

Generates a Orange.Span struct

  
    
  
  Options


	:style - style attributes for the line. Supported keys are:
	:color - see Color section

	:background_color - see Color section





  
    
  
  Children validation


Span only accepts a single string as children

  
    
  
  Examples


iex> import Orange.Macro
iex> span style: [color: :red] do
...>   "foo"
...> end
%Orange.Span{children: ["foo"], attributes: [style: [color: :red]]}

  


        

      



  

    
Orange.Terminal behaviour
    



      
Provides API to interact with the terminal.

      


      
        Summary


  
    Callbacks
  


    
      
        disable_raw_mode()

      


    


    
      
        draw(buffer)

      


    


    
      
        draw(buffer, previous_buffer)

      


    


    
      
        enable_raw_mode()

      


    


    
      
        enter_alternate_screen()

      


    


    
      
        hide_cursor()

      


    


    
      
        leave_alternate_screen()

      


    


    
      
        poll_event()

      


    


    
      
        show_cursor()

      


    


    
      
        terminal_size()

      


    





  
    Functions
  


    
      
        disable_raw_mode()

      


        Disable terminal raw mode.



    


    
      
        draw(buffer, previous_buffer \\ nil)

      


        Draws the buffer to the terminal. If a previous buffer is provided, it will only draw the diff between the two buffers.



    


    
      
        enable_raw_mode()

      


        Enable terminal raw mode.



    


    
      
        enter_alternate_screen()

      


        Enter terminal alternate screen.



    


    
      
        hide_cursor()

      


        Hide terminal cursor.



    


    
      
        leave_alternate_screen()

      


        Leave terminal alternate screen.



    


    
      
        poll_event()

      


        Polls the terminal for events.



    


    
      
        show_cursor()

      


        Show terminal cursor.



    


    
      
        terminal_size()

      


        See Orange.Terminal.Binding.terminal_size/0.



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    disable_raw_mode()



  


  

      

          @callback disable_raw_mode() :: :ok


      



  



  
    
      
      Link to this callback
    
    draw(buffer)



  


  

      

          @callback draw(buffer :: Orange.Renderer.Buffer.t()) :: :ok


      



  



  
    
      
      Link to this callback
    
    draw(buffer, previous_buffer)



  


  

      

          @callback draw(
  buffer :: Orange.Renderer.Buffer.t(),
  previous_buffer :: Orange.Renderer.Buffer.t()
) :: :ok


      



  



  
    
      
      Link to this callback
    
    enable_raw_mode()



  


  

      

          @callback enable_raw_mode() :: :ok


      



  



  
    
      
      Link to this callback
    
    enter_alternate_screen()



  


  

      

          @callback enter_alternate_screen() :: :ok


      



  



  
    
      
      Link to this callback
    
    hide_cursor()



  


  

      

          @callback hide_cursor() :: :ok


      



  



  
    
      
      Link to this callback
    
    leave_alternate_screen()



  


  

      

          @callback leave_alternate_screen() :: :ok


      



  



  
    
      
      Link to this callback
    
    poll_event()



  


  

      

          @callback poll_event() :: [Orange.Terminal.KeyEvent.t()]


      



  



  
    
      
      Link to this callback
    
    show_cursor()



  


  

      

          @callback show_cursor() :: :ok


      



  



  
    
      
      Link to this callback
    
    terminal_size()



  


  

      

          @callback terminal_size() :: {non_neg_integer(), non_neg_integer()}


      



  


        

      

      
        Functions

        


  
    
      
      Link to this function
    
    disable_raw_mode()



  


  

Disable terminal raw mode.

  



    

  
    
      
      Link to this function
    
    draw(buffer, previous_buffer \\ nil)



  


  

Draws the buffer to the terminal. If a previous buffer is provided, it will only draw the diff between the two buffers.

  



  
    
      
      Link to this function
    
    enable_raw_mode()



  


  

Enable terminal raw mode.

  



  
    
      
      Link to this function
    
    enter_alternate_screen()



  


  

Enter terminal alternate screen.

  



  
    
      
      Link to this function
    
    hide_cursor()



  


  

Hide terminal cursor.

  



  
    
      
      Link to this function
    
    leave_alternate_screen()



  


  

Leave terminal alternate screen.

  



  
    
      
      Link to this function
    
    poll_event()



  


  

Polls the terminal for events.

  



  
    
      
      Link to this function
    
    show_cursor()



  


  

Show terminal cursor.

  



  
    
      
      Link to this function
    
    terminal_size()



  


  

See Orange.Terminal.Binding.terminal_size/0.

  


        

      



  

    
Orange.Component behaviour
    



      
A behaviour module for implementing custom components.
A custom component is a self-contained UI component that can be reused across the application. Custom components are composed of other custom or primitive components.
Custom components can have internal state and receive terminal events to update their state. They also have hooks, which will be trigger at specific points in the component lifecycle, for example, after_mount/1 and after_unmount/1

  
    
  
  Examples


In order to implement a custom component, you need to define a module that implements the Orange.Component behaviour. As an example, we will implement a Counter component. We can use up and down button to increase or decrease the counter value.
defmodule Counter do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: 0, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: :up} ->
        state + 1

      %Orange.Terminal.KeyEvent{code: :down} ->
        state - 1

      _ ->
        state
    end
  end

  @impl true
  def after_mount(state, _attrs, update) do
    # Set counter value to 10 after 5 seconds
    spawn(fn ->
      Process.sleep(5000)
      update.(10)
    end)
  end

  @impl true
  def render(state, attrs, _update) do
    rect style: [width: "100%", height: "1fr", border: attrs[:highlight]] do
      span do
        "Counter: #{state}"
      end
    end
  end
end
This component can be used as children for other components like this:
defmodule App do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil, events_subscription: true}

  @impl true
  def handle_event(event, state, _attrs) do
    case event do
      %Orange.Terminal.KeyEvent{code: {:char, "q"}} ->
        Orange.stop()
        state

      _ ->
        state
    end
  end

  def render(_state, _attrs, _update) do
    rect style: [width: 20, height: 20, border: true] do
      # No attributes
      Counter

      # Or with attributes
      {Counter, id: :counter, highlight: true}
    end
  end
end
[image: Rendered result]

  
    
  
  Update callback


There are two ways to update the component state:
	By returning a new state from the handle_event/3 callback

	By using the update callback


The update callback will be passed to the render/3 and lifecycle hooks. The update callback can be called with a new state or a function that receives the current state and returns the new state.
def render(state, attrs, update_callback)
  update_callback.(%{state | counter: state.counter + 1})
  update_callback.(fn state -> %{state | counter: state.counter - 1} end)
end

      


      
        Summary


  
    Types
  


    
      
        event()

      


    


    
      
        state()

      


    


    
      
        ui_element()

      


    


    
      
        update_callback()

      


    





  
    Callbacks
  


    
      
        after_mount(state, attributes, update_callback)

      


        Lifecycle hook that is called after the component is mounted (first time render to the terminal).



    


    
      
        after_unmount(state, attributes, update_callback)

      


        Lifecycle hook that is called after the component is unmounted (remove from the terminal).



    


    
      
        handle_event(event, state, attributes)

      


        Receives the event, current state, and attributes and returns the new state.



    


    
      
        init(attributes)

      


        Receives the attributes map from parent component and returns the component configuration.



    


    
      
        render(state, attributes, update_callback)

      


        Receives the current state, attributes and update callback and returns the rendered UI element.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    event()



  


  

      

          @type event() :: Orange.Terminal.KeyEvent.t()


      



  



  
    
      
      Link to this type
    
    state()



  


  

      

          @type state() :: map()


      



  



  
    
      
      Link to this type
    
    ui_element()



  


  

      

          @type ui_element() :: Orange.Rect.t() | Orange.Line.t() | Orange.Span.t()


      



  



  
    
      
      Link to this type
    
    update_callback()



  


  

      

          @type update_callback() :: (state() -> state()) | state()


      



  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    after_mount(state, attributes, update_callback)



      (optional)

  


  

      

          @callback after_mount(state(), attributes :: map(), update_callback()) :: any()


      


Lifecycle hook that is called after the component is mounted (first time render to the terminal).

  



  
    
      
      Link to this callback
    
    after_unmount(state, attributes, update_callback)



      (optional)

  


  

      

          @callback after_unmount(state(), attributes :: map(), update_callback()) :: any()


      


Lifecycle hook that is called after the component is unmounted (remove from the terminal).

  



  
    
      
      Link to this callback
    
    handle_event(event, state, attributes)



      (optional)

  


  

      

          @callback handle_event(event(), state(), attributes :: map()) :: state()


      


Receives the event, current state, and attributes and returns the new state.

  



  
    
      
      Link to this callback
    
    init(attributes)



  


  

      

          @callback init(attributes :: map()) :: %{state: state(), events_subscription: boolean()}


      


Receives the attributes map from parent component and returns the component configuration.
Configuration keys are:
	state - The initial state of the component

	events_subscription - whether the component should subscribe to terminal events



  



  
    
      
      Link to this callback
    
    render(state, attributes, update_callback)



  


  

      

          @callback render(state(), attributes :: map(), update_callback()) :: ui_element()


      


Receives the current state, attributes and update callback and returns the rendered UI element.
Info
This function must either return a single component or nil.


  


        

      



  

    
Orange.Component.Input 
    



      
An uncontrolled input component.

  
    
  
  Attributes


	:on_submit - A callback triggered when input is submitted. This attribute is required.

	:submit_key - The keyboard key that will trigger submission. This attribute is optional and defaults to :enter. See Orange.Terminal.KeyEvent for supported values.

	:on_exit - A callback triggered when input is exited. This attribute is optional.

	:exit_key - The keyboard key that will unfocus the input. This attribute is optional and defaults to :esc. See Orange.Terminal.KeyEvent for supported values.
Info
The :submit_key and :exit_key can not be :backspace as it is reserved for deleting characters.


	:auto_focus - Whether to focus automatically after mount. This attribute is optional. If true, the :id attribute is required and the input will unfocus after submission.

	:prefix - The input prefix string. This attribute is optional.

	:style - The component style. This attribute is optional.



  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: %{search_value: ""}}

  @impl true
  def render(state, _attrs, update) do
    rect style: [width: 50, height: 20] do
      {
        Orange.Component.Input,
        id: :input,
        on_submit: fn value -> update.(%{state | search_value: value}) end,
        submit_key: {:char, "x"},
        prefix: "Enter search value:",
        auto_focus: true
      }

      line do
        "Search value: #{state.search_value}"
      end
    end
  end
end
  [image: rendered result]

      





  

    
Orange.Component.Modal 
    



      
An modal/dialog component which renders an overlay in the middle of the viewport.

  
    
  
  Attributes


	:open - Whether the modal is open or not. This attribute is required.

	:offset_x - The offset from the left and right edge of the screen. This attribute is required.

	:offset_y - The offset from the top and bottom edge of the screen. This attribute is required.
Info
When the offset_x/y is too big for the terminal size, the modal will not be rendered.


	:children - The content of the modal. This attribute is optional.



  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: %{search_value: ""}}

  @impl true
  def render(state, _attrs, update) do
    modal_content =
      rect do
        "foo"
        "bar"
      end

    rect do
      line do
        "Displaying modal..."
      end

      {
        Orange.Component.Modal,
        offset_x: 8,
        offset_y: 4,
        children: modal_content,
        open: true
      }
    end
  end
end
  [image: rendered result]

      





  

    
Orange.Component.VerticalScrollableRect 
    



      
A component that renders a vertical scrollable rectangle. When the container is overflown, a scrollbar is rendered.

  
    
  
  Attributes


	:height - The height of the container rect. This attribute is required.
	:content_height - The height of the inner content. This attribute is required.
	:scroll_offset - The vertical scroll offset. This attribute is required.
	:children - The children of the container rect. This attribute is required.
	:title - The title of the scrolling box. This attribute is optional.


  
    
  
  Examples


defmodule Example do
  @behaviour Orange.Component

  import Orange.Macro

  @impl true
  def init(_attrs), do: %{state: nil}

  @impl true
  def render(_state, attrs, _update) do
    total_items = 30

    rect style: [width: 20, height: 20, border: attrs[:highlight]] do
      {
        Orange.Component.VerticalScrollableRect,
        height: 20,
        content_height: total_items,
        scroll_offset: 5,
        children: for(i <- 1..total_items, do: "Item #{i}")
      }
    end
  end
end
  [image: rendered result]
  For more examples, see examples/scrollable_rect.exs.

      





  

    
Orange.Terminal.KeyEvent 
    



      
Terminal keyboard events.
See key_code/0 below for supported keys.

      


      
        Summary


  
    Types
  


    
      
        key_code()

      


    


    
      
        key_modifier()

      


    


    
      
        t()

      


    





      


      
        Types

        


  
    
      
      Link to this type
    
    key_code()



  


  

      

          @type key_code() ::
  :backspace
  | :enter
  | :left
  | :right
  | :up
  | :down
  | :home
  | :end
  | :page_up
  | :page_down
  | :tab
  | :back_tab
  | :delete
  | :insert
  | :f1
  | :f2
  | :f3
  | :f4
  | :f5
  | :f6
  | :f7
  | :f8
  | :f9
  | :f10
  | :f11
  | :f12
  | :null
  | :esc
  | :caps_lock
  | :scroll_lock
  | :num_lock
  | :print_screen
  | :pause
  | :menu
  | :keypad_begin
  | {:char, String.t()}


      



  



  
    
      
      Link to this type
    
    key_modifier()



  


  

      

          @type key_modifier() :: :shift | :ctrl | :alt | :super | :hyper


      



  



  
    
      
      Link to this type
    
    t()



  


  

      

          @type t() :: %Orange.Terminal.KeyEvent{code: key_code(), modifiers: [key_modifier()]}

